


9/7.8



Minis jeux pour programmeur pressé

Les jeux les plus longs ne sont pas toujours les meilleurs. Certains adeptes du joystick diront même que les jeux les plus chers ne sont pas toujours les meilleurs.

Partant de la première considération, nous avons décidé de vous proposer des jeux que nous nommerons minis, car leur listing ne vous demanderont pas des heures de frappe, et auront toutes les chances de vous délasser après quelques heures d'utilisation de votre tableur ou votre gestion de compte préférée.

Pour ne frustrer personne, nous vous proposerons aussi bien des jeux de réflexion, de chance, que d'Arcade.

De plus, par leur simplicité d'utilisation, ces jeux pourront être utilisés par les moins jeunes, ce qui pourra parfaire leur esprit de raisonnement.

Bien sûr, ne nous demandez pas l'impossible : vous ne trouverez pas, étant donné la taille des programmes et le langage utilisé, de SUPER STARTREK avec vue en trois dimensions des vaisseaux ennemis et autres torpilles à photons multicolores.

UN MENU POUR CINQ MINIS JEUX

Les lecteurs pressés de jouer pourront directement passer à la frappe des jeux proposés plus loin ; mais nous pensons qu'ils manqueront quelque chose en n'utilisant pas le premier programme que nous leur proposons ici.

Ce programme consiste à présenter sous forme de menu les différents jeux qui pourront être utilisés à partir de celui-ci.

Par ce menu, nous proposons cinq choix de jeux, qui sont numérotés de 1 à 5. Le simple appui sur une touche de sélection lance directement un jeu parmi ceux que vous y aurez entré.

Voyons tout de suite le listing de celui-ci, pour ensuite l'étudier et repérer la façon de lui adjoindre des minis jeux.

```
10 REM *****
20 REM ** **
30 REM ** MENU POUR MINIS JEUX **
40 REM ** **
50 REM *****
60 REM
70 ON BREAK GOSUB 1000
80 MODE 2
90 SPEED KEY 19,3
100 CALL &BB4E
110 REM
120 PLOT 150,150
130 DRAW 150,380
140 DRAW 500,380
150 DRAW 500,150
160 DRAW 150,150
170 LOCATE 35,23
180 PRINT "MINI JEUX"
190 LOCATE 25,5
200 PRINT "1 -> Jeu numero 1"
210 LOCATE 25,7
220 PRINT "2 -> Jeu numero 2"
230 LOCATE 25,9
240 PRINT "3 -> Jeu numero 3"
250 LOCATE 25,11
260 PRINT "4 -> Jeu numero 4"
270 LOCATE 25,13
280 PRINT "5 -> Jeu numero 5"
290 LOCATE 29,20
300 PRINT "FAITES VOTRE CHOIX"
310 WHILE INKEY# <> ""
320 WEND
330 WHILE a# <> "1" AND a# <> "2" AND a#
   <> "3" AND a# <> "4" AND a# <> "5"
340     FLAG% = 0
350     a# = ""
360     WHILE a# = ""
370         a# = INKEY#
380         LOCATE 25,5
390         PRINT CHR$(24); "1"
400         LOCATE 25,7
410         PRINT CHR$(24); "2"
420         LOCATE 25,9
430         PRINT CHR$(24); "3"
440         LOCATE 25,11
450         PRINT CHR$(24); "4"
460         LOCATE 25,13
```

```
470         PRINT CHR$(24);"5"
480         FOR I% = 1 TO 200
490         NEXT I%
500         FLAG% = FLAG% + 1
510         FLAG% = FLAG% MOD 2
520     WEND
530 WEND
540 CALL &BB4E
550 LOCATE 29,20
560 PRINT "Ok, on y va ....."
570 a = VAL(a$)
580 ON a GOSUB 3000,5000,7000,9000,11000
590 RUN
980 '
990 '
1000 MODE 2
1010 PRINT "Ok, j'arrete le jeu"
1020 PRINT
1070 PRINT "Frappez sur <ESPACE> pour re
lancer le menu"
1080 PRINT "sinon sur <S>"
1090 CALL &BB4E
1100 WHILE INKEY$ <> ""
1110 WEND
1120 a$ = ""
1130 WHILE a$ = ""
1140 a$ = INKEY$
1150 WEND
1160 IF a$ = CHR$(32) THEN RUN
1170 SPEED KEY 19,3
1180 STOP
2980 '
2990 '
3000 MODE 2
3010 PRINT "Jeu numero 1"
3020 FOR i% = 1 TO 3000
3030 NEXT i%
3040 RETURN
4980 '
4990 '
5000 MODE 2
5010 PRINT "Jeu numero 2"
5020 FOR i% = 1 TO 3000
5030 NEXT i%
5040 RETURN
6980 '
6990 '
```

```
7000 MODE 2
7010 PRINT "Jeu numero 3"
7020 FOR i% = 1 TO 3000
7030 NEXT i%
7040 RETURN
8980 '
8990 '
9000 MODE 2
9010 PRINT "Jeu numero 4"
9020 FOR i% = 1 TO 3000
9030 NEXT i%
9040 RETURN
10980 '
10990 '
11000 MODE 2
11010 PRINT "Jeu numero 5"
11020 FOR i% = 1 TO 3000
11030 NEXT i%
11040 RETURN
```

Avec la présentation algorithmique adoptée, il est très simple de comprendre le fonctionnement de celui-ci.

En ligne 70, on interdit l'arrêt du programme lors de deux appuis sur <ESC>. Cette fonction est détournée vers la ligne 1000, que nous étudierons plus loin.

La ligne 90 rétablit une vitesse d'acquisition des touches convenables, car celle-ci est modifiée par certains des programmes que nous présentons.

La ligne 100 rétablit la police de caractères originale, car certains de nos programmes modifient cette police par l'utilisation des instructions **SYMBOL AFTER** et **SYMBOL**.

Le menu est présenté entre les lignes 120 et 300. Vous remarquerez que pour l'instant, le nom des jeux n'est pas encore proposé. Cela sera fait au moment de la présentation des minis jeux.

Lignes 310 et 320 : on vide le buffer du clavier par une boucle de lecture de type **TANTQUE ... FINTANTQUE**.

Ligne 330 : ne sont prises en compte que les touches correspondantes aux jeux proposés. Si vous désirez proposer un sixième jeu, vous devrez ajouter à cette ligne les instructions :

```
AND a$ <> "6"
```

Lignes 360 à 520 : on attend la frappe d'une touche tout en effectuant une animation sur les numéros des minis jeux. Cette animation, très simple, consiste à réécrire les numéros de jeux en vidéo inversée l'un après l'autre.

Ligne 540 : annulation vidéo inversée.

Lignes 560 à 580 : aiguillage vers le programme choisi, dont le début se trouve aux lignes suivantes :

Programme 1 → ligne 3000

Programme 2 → ligne 5000

Programme 3 → ligne 7000

Programme 4 → ligne 9000

Programme 5 → ligne 11000

Ici encore, vous aurez à ajouter le numéro de ligne du début de votre programme si vous en ajoutez un sixième.

Lignes 1000 à 1180 : traitement de l'interruption du programme. Le CPC signale qu'il arrête le jeu en cours, et propose un accès au menu, si l'on frappe sur la touche <ESPACE>, sinon, il redonne la main au Basic.

Si vous ne désirez pas redonner la main à Basic, remplacez la ligne 1100 par la ligne suivante :

```
1100 RUN
```

Les lignes suivantes correspondent aux lignes de traitement des différents minis jeux, dans lesquelles nous avons pour l'instant proposé l'affichage du numéro de jeu sélectionné, pour vous permettre d'essayer ce menu.

Un conseil, avant de lancer ce programme, sauvegardez-le sur votre disquette, surtout si vous avez modifié la ligne 1100, car après lancement vous risquez de ne plus avoir le retour à Basic autrement que par l'appui sur <SHIFT> <CONTROL> et <ESC>.

A LA RECHERCHE DU SAVON PERDU

Lors de votre dernier bain, ne vous souvenez-vous pas avoir perdu votre savon au fond de votre baignoire hydromassante dernier cri ? Avec toutes ces bulles et la mousse du bain moussant, impossible de localiser ce maudit savon.

Ah ! Si la baignoire pouvait parler !

Le principe du jeu

Donner la parole à une baignoire ? Il suffit de demander : WEKA vous offre dans ce mini jeu une baignoire qui parle afin que vous localisiez facilement le morceau de savon qui vous permettra de continuer votre toilette sans risque de débordement d'injures.

Notre baignoire est représentée par un carré de 10 sur 10 (tant pis pour les grands, ils n'ont qu'à prendre des douches).

Vous indiquez dans ce carré une case où vous désirez commencer votre recherche, et la baignoire vous indique, par affichage sur l'écran, la direction dans laquelle il faut continuer.

Mais, notre baignoire est un peu facétieuse, car un ancien marin, lors d'un débarbouillage y a perdu une boussole qu'elle s'est empressée d'avaler. Aussi, ses indications sont-elles données en rapport avec la rose des vents (cherchez un peu plus au Nord, ou à l'Est, ou au Sud-Ouest...).

Vous devrez donc vous aligner sur le nord magnétique si vous voulez retrouver rapidement votre savon, car au bout de cinq essais, la baignoire se vide, vous faisant découvrir son emplacement.

Le programme

Pour intégrer ce premier mini jeu au menu, il vous suffira de modifier la ligne 200 du programme de menu en :

200 PRINT "1 → A la recherche du savon"

Afin de répondre à l'appel correspondant, nous avons débuté notre programme à la ligne 2960, la ligne 3000 devenant la ligne effective de traitement lors de l'appel.

```
2960 REM *****
2961 REM *
2970 REM * RECHERCHEZ LE SAVON *
2971 REM *
2980 REM *****
2990 REM
3000 CALL &BB4E: REM annule SYMBOL AFTER
3010 MODE 1
3020 BORDER 9
3030 INK 0,0:INK 1,13:INK 2,1:INK 3,12
3040 WINDOW #0,1,40,16,25
3050 PAPER 0 : PEN 1
3060 CLS
3070 WINDOW #1,12,25,2,12
3080 PAPER #1,2 : PEN #1,3
3090 CLS #1
3100 GRAPHICS PEN 1
3110 MOVE 400,206
3120 DRAW 400,384
3130 DRAW 174,384
3140 DRAW 174,206
3150 DRAW 400,206
3160 LOCATE #1,1,12
3170 FOR I = 10 TO 1 STEP -1
3180     I$ = RIGHT$(STR$(I),2)
3190     PRINT#1,I$
3200 NEXT I
3210 PRINT#1,STRING$(2,CHR$(&20));
3220 FOR I = 1 TO 9
3230     I$ = RIGHT$(STR$(I),1)
3240     PRINT#1,I$;
3250 NEXT I
3260 PRINT#1,"10"
3270 GRAPHICS PEN 3
3280 MOVE 210,228
3290 DRAW 382,228
3300 DRAW 382,380
3310 DRAW 210,380
3320 DRAW 210,228
3330 LOCATE #1,8,3
3340 PRINT#1,"N" + CHR$(8) + CHR$(10) +
CHR$(240);
3350 PRINT#1,CHR$(8) + CHR$(8) + CHR$(8)
+ CHR$(10);
3360 PRINT#1,"D" + CHR$(242) + " " + CHR
$(243) + "E";
3370 PRINT#1,CHR$(8) + CHR$(8) + CHR$(8)
+ CHR$(10);
```

```
3380 PRINT#1,CHR$(241) + CHR$(8) + CHR$(
10) + "S"
3390 LOCATE #1,4,7
3400 PRINT#1,"y" + CHR$(8)+CHR$(10);
3410 PRINT#1,CHR$(240) + CHR$(10);
3420 PRINT#1,CHR$(243) + "x"
3430 REM
3440 REM *** REGLE DU JEU ***
3450 REM
3460 CLS
3470 PRINT "Vous avez perdu votre savon
dans votre"
3480 PRINT "baignoire et devez le retrou
ver a tatons"
3490 PRINT
3500 PRINT "Appuyez sur une touche pour
la suite"
3510 CALL &BB06
3520 CLS
3530 REM
3540 REM *** DEBUT DU JEU ***
3550 REM
3560 CLS
3570 REM REM * INITIALISATINS *
3580 nbessai% = 5 : REM Nombre maximum d
'essais
3590 REM pour un jeu optimal
3600 REM prendre nbessai% = 4
3610 positionx = INT(RND * 10) + 1: REM
ordonnee initiale
3620 positiony = INT(RND * 10) + 1: REM
abscisse initiale
3630 REM
3640 REM * essais *
3650 REM
3660 FOR essai% = 1 TO nbessai%
3670     PRINT STRING$(10,CHR$(32));
3680     PRINT "Essai numero";
3690     PRINT essai%
3700     PRINT "Entrez les coordonnees (x
,y)
3710     INPUT abscisse,ordonnee
3720     PRINT
3730     IF ABS(abscisse - positionx) + A
BS(ordonnee - positiony) = 0 THEN 4100 :
    REM gagne?
3740     GOSUB 3940: REM aide de direction
```

```
3750 PRINT
3760 NEXT essai%
3770 REM
3780 REM * plus d'essais *
3790 REM
3800 CLS
3810 PRINT "Desole vous avez perdu,"
3820 PRINT "Le savon se trouvait en";
3830 PRINT STRING$(2,CHR$(32));
3840 PRINT positionx;" , ";positiony
3850 PRINT
3860 PRINT "Appuyez sur une touche"
3870 CALL &BB06
3880 CLS
3890 PRINT "Une autre partie"
3900 GOTO 3610
3910 REM
3920 REM * aide de direction *
3930 REM
3940 PRINT "Cherchez plutot en direction
:"
3950 IF ordonnee = positiony THEN 4000
3960 IF ordonnee < positiony THEN 3990
3970 PRINT "Sud ";
3980 GOTO 4000
3990 PRINT "Nord ";
4000 IF abscisse = positionx THEN 4050
4010 IF abscisse < positionx THEN 4040
4020 PRINT "Ouest"
4030 GOTO 4050
4040 PRINT "Est";
4050 PRINT
4060 RETURN
4070 REM
4080 REM * savon trouve *
4090 REM
4100 CLS
4110 PRINT
4120 PRINT "Bravo, vous avez trouve le s
avon,"
4130 PRINT "en";
4140 PRINT essai%;
4150 PRINT " essais"
4160 PRINT
4170 PRINT "Appuyez sur une touche"
4180 CALL &BB06
4190 CLS
4200 GOTO 3890
```

Lignes 3000 à 3090 : initialisation du jeu de caractères standard, des couleurs d'encres utilisées et des fenêtres de dialogues.

Lignes 3100 à 3420 : dessin de la baignoire et rappel de la rose des vents, ainsi que des axes de la baignoire.

Lignes 3460 à 3510 : énonce bref du principe du jeu.

Lignes 3580 à 3620 : initialisation du nombre maximal d'essais et recherche d'une position aléatoire du savon dans la baignoire.

Lignes 3660 à 3760 : boucle principale du jeu permettant d'acquérir les coordonnées de recherche. On sort de celle-ci si le savon est trouvé, ou si le nombre d'essais maximal est atteint. Sinon, la baignoire propose une aide en exécutant le sous-programme de la ligne 3940.

Lignes 3800 à 3900 : le nombre d'essais maximal étant atteint, le programme le signale et indique ici la position du savon.

Lignes 3940 à 4060 : le programme compare la position de recherche et celle du savon pour fournir une direction pour continuer la recherche.

Lignes 4100 à 4200 : programme de traitement lorsque le savon a été trouvé.

AU VOLANT D'UNE F1

Le jeu

Après le choix 2 du menu, vous voici parachuté au volant d'une formule 1 en pleine campagne (on ne veut plus de vous sur les pistes, vous êtes trop dangereux) pour y effectuer votre entraînement.

La chance est tout de même avec vous, on vous laisse choisir la largeur de la route entre 5 et 20 mètres.

Attention, une fois ce choix effectué, vous voici lancé sur une route bordée d'arbres qui auront toujours une fâcheuse tendance à se rapprocher de votre voiture.

Heureusement, après un entraînement assidu à diriger votre voiture à l'aide des flèches droite et gauche, vous en viendrez vite aux routes de 5 mètres de large et pourrez inscrire votre nom au palmarès des meilleurs conducteurs qui s'affiche en haut de l'écran.


```
5260 score% = 0
5270 LOCATE 20,10
5280 PRINT "Haut score :";
5290 PRINT scorehaut%
5300 LOCATE 23,11
5310 PRINT "(" + nom$ + ")"
5320 LOCATE 1,10
5330 PRINT "Score : "
5340 routegauche% = 20 - INT(largeur% /
2)
5350 position% = 20
5360 LOCATE routegauche%,20
5370 PRINT arbre$
5380 routedroite% = routegauche% + large
ur%
5390 LOCATE routedroite%,20
5400 PRINT arbre$
5410 LOCATE position%,21
5420 PRINT voiture$
5430 LOCATE 10,5
5440 PRINT "Appuyez sur une touche"
5450 CALL &BBO6
5460 REM
5470 REM ** LANCEMENT COURSE **
5480 REM
5490 LOCATE 10,5
5500 PRINT STRING$(24,CHR$(32))
5510 score% = 0
5520 EVERY 50,1 GOSUB 6090 : REM afficha
ge score
5530 DI
5540 LOCATE routegauche%,20
5550 PRINT arbre$
5560 routedroite% = routegauche% + large
ur%
5570 LOCATE routedroite%,20
5580 PRINT arbre$
5590 LOCATE position%,21
5600 PRINT voiture$
5610 IF position% = routegauche% + 1 THE
N 5810
5620 IF position% = routedroite% - 1 THE
N 5810
5630 a$ = ""
5640 a$ = INKEY$
5650 score% = score% + 1
```

```
5660 EI
5670 IF a$ = "" THEN GOTO 5720
5680 IF a$ = CHR$(242) THEN position% =
position% - 1
5690 IF a$ = CHR$(243) THEN position% =
position% + 1
5700 IF position% < routegauche% THEN po
sition% = position% + 1:GOTO 5810
5710 IF position% > routedroite% THEN po
sition% = position% - 1:GOTO 5810
5720 hasard = RND
5730 IF hasard > 0.62 THEN routegauche%
= routegauche% - 1
5740 IF hasard < 0.38 THEN routegauche%
= routegauche% + 1
5750 IF routegauche% < 1 THEN routegauch
e% = 1
5760 IF routegauche% + largeur% > 38 THE
N routegauche% = routegauche% - 1
5770 GOTO 5530
5780 REM
5790 REM ** ACCIDENT **
5800 REM
5810 SPEED KEY 19,3
5820 LOCATE position%,21
5830 PRINT crash1$
5840 FOR i% = 1 TO 1000 : NEXT i%
5850 LOCATE position%,21
5860 PRINT crash2$
5870 FOR i% = 1 TO 1000 : NEXT i%
5880 LOCATE position%,21
5890 PRINT crash1$
5900 FOR i% = 1 TO 1000 : NEXT i%
5910 LOCATE position%,21
5920 PRINT crash3$
5930 FOR i% = 1 TO 1000 : NEXT i%
5940 PRINT
5950 IF scorehaut% >= score% THEN GOTO 6
010
5960 scorehaut% = score%
5970 SPEED KEY 19,3
5980 PRINT "Bravo, vous avez battu le re
cord"
5990 INPUT "Votre Nom :";nom$
6000 a$ = "A"
6010 FOR i% = 1 TO 500
6020     a$ = INKEY$
```

```

6030 NEXT i%
6040 GOTO 5250
6050 REM
6060 REM * affichage periodique *
6070 REM *      du score      *
6080 REM
6090 LOCATE 10,10
6100 PRINT score%
6110 RETURN
6120 REM *****
6130 END

```

Ligne 5010 : retour au jeu de caractères standard.

Ligne 5050 : les caractères à partir de 250 vont être redéfinis.

Lignes 5060 à 5080 : création du dessin d'un arbre qui est rangé dans la variable `arbre$`.

Lignes 5090 et 5100 : création du dessin de la voiture, rangée dans la variable `voiture$`.

Lignes 5110 à 5160 : trois dessins de la voiture se décomposant sous l'action d'un choc violent sur un arbre.

Ligne 5220 : la répétabilité des touches est placée à une vitesse correcte pour permettre l'acquisition de la largeur de la route en ligne 5230.

Ligne 5240 : la répétabilité des touches est accélérée pour rendre le déplacement de la voiture plus rapide (instruction `SPEED KEY`).

Lignes 5250 à 5330 : affichage des scores.

Lignes 5320 à 5450 : affichage de la ligne de départ avec calcul du positionnement de la voiture selon la largeur de la route.

Lignes 5490 à 5770 : gestion de la course. Les touches du clavier sont scrutées pour détecter les déplacements de la voiture. Les courbes de la route sont déterminées aléatoirement et permettent ainsi un rapprochement/éloignement des arbres. On gère aussi en lignes 5750 et 5760 les éventuels débordements de la route en dehors des limites de l'écran, ce qui entraînerait un pilotage sans visibilité, sur une route à deux niveaux (essayez d'éliminer la ligne 5750).

Lignes 5810 à 5940 : traitement d'une sortie de route par affichage des débris de la voiture.

Lignes 5950 à 6040 : acquisition du nom, si votre performance est la meilleure.

Lignes 6890 à 6110 : sous-programme d'affichage périodique du score qui est initialisé en ligne 5520.

UN MORPION ÉVOLUTIF

Principe du jeu

Tout collégien a joué un jour au morpion sur le bancs de l'école, lors d'un cours un peu soporifique. Chaque joueur marque une croix sur une grille et doit réaliser un alignement de 4, 5 ou 6 croix dans n'importe quelle direction, selon un nombre fixé d'avance.

Nous avons décidé de modifier cette règle pour vous permettre de jouer contre l'ordinateur, et de découvrir les perspectives d'évolution de ce jeu.

Le but sera, sur une grille de 15 cases sur 15 de créer un alignement maximal de pions, l'ordinateur s'efforçant de vous contrer. Bien sûr, nous n'avons pas réalisé un programme trop performant, pour ne pas lasser le joueur, celui-ci gardant une marge d'erreur aléatoire (ce qui a permis à l'auteur d'aligner 12 de ses pions, ouf !).

Lors du jeu, si sa recherche est trop longue, l'ordinateur se garde la possibilité d'abandonner la partie et vous êtes dans ce cas déclaré gagnant.

Vous pouvez aussi déclarer forfait en entrant les coordonnées 0,0.

Le programme

Pour implanter ce programme, il vous faudra modifier la ligne 240 du programme de menu en :

240 PRINT "3 → Le morpion evolutif"

Le programme débute en 6940 et possède sa première ligne exécutable au numéro d'appel 7000.

```

6940 REM *****
6950 REM **                               **
6960 REM **   MORPION MAXIMUM           **
6970 REM **                               **
6980 REM *****
6990 REM
7000 RANDOMIZE TIME
7010 MODE 2
7020 PRINT TAB(20);"LE MORPION MAXIMUM"
7030 PRINT:PRINT
7040 PRINT "Materiel : Une grille de 15
X 15 numerote dans l'ordre suivant :"
7050 PRINT "   i 2 3 4 5 6 ...   -->
colonnes"
7060 FOR i% = 1 TO 6
7070     PRINT i%;
7080     FOR J% = 1 TO 6
7090         PRINT " .";
7100     NEXT J%
7110     PRINT
7120 NEXT i%
7130 FOR i% = 1 TO 3
7140     PRINT " ."
7150 NEXT i%
7160 PRINT " lignes"
7170 PRINT
7180 PRINT " Vous devez aligner le maxim
um de pions, dans toutes les"
7190 PRINT " directions possibles (verti
cale, horizontale, diagonale)"
7200 PRINT
7210 PRINT " Votre pion : ";
7220 PRINT CHR$(231)
7230 PRINT " Le CPC      : ";
7240 PRINT CHR$(232)
7250 PRINT "Vous entrez la position du p
ion a placer dans l'ordre"
7260 PRINT TAB(30);"colonne,ligne"
7270 PRINT
7280 PRINT TAB(20);" <APPUYEZ SUR UNE TO
UCHE POUR JOUER>"
7290 CALL &BB06
7300 MODE 2
7310 WINDOW #1,1,80,17,25
7320 REM
7330 REM ** INITIALISATION TABLEAU **
7340 DIM tableau(15,15)

```

```
7350 FOR i = 1 TO 15
7360   FOR j = 1 TO 15
7370     tableau(i,j) = 46
7380   NEXT j
7390 NEXT i
7400 REM
7410 REM ** AFFICHAGE TABLEAU **
7420 REM
7430 LOCATE 1,1
7440 FOR i = 1 TO 15
7450   FOR j = 1 TO 15
7460     LOCATE 3*i,j
7470     PRINT CHR$(tableau(i,j))
7480   NEXT j
7490   PRINT
7500 NEXT i
7510 PRINT
7520 REM
7530 REM ** ATTENTE JEU DU JOUEUR **
7540 REM
7550 PRINT #1, "Vous commencez .."
7560 PRINT #1, "Votre jeu (ligne,colonne
) ";
7570 INPUT #1, i,j
7580 REM
7590 REM ** fin du jeu? **
7600 REM
7610 IF i = 0 THEN GOTO 8820
7620 x = i
7630 y = j
7640 REM
7650 REM ** position autorisee? **
7660 REM
7670 flag = 1
7680 IF x<1 THEN GOTO 7730
7690 IF x > 15 THEN GOTO 7730
7700 IF y < 1 THEN GOTO 7730
7710 IF y > 15 THEN GOTO 7730
7720 GOTO 7740
7730 flag = 0
7740 IF flag = 1 THEN 7800
7750 PRINT #1, "position illegale..."
7760 GOTO 7560
7770 REM
7780 REM ** case libre ? **
7790 REM
7800 IF tableau(i,j) = 46 THEN GOTO 7860
7810 PRINT #1, "Case occupe"
```

```
7820 GOTO 7560
7830 REM
7840 REM ** place le pion **
7850 REM
7860 tableau (i,j) = 230
7870 REM
7880 REM ** affiche le pion **
7890 REM
7900 LOCATE 3*i,j
7910 PRINT CHR$(tableau(i,j))
7920 REM
7930 REM ** ESSAI ORDINATEUR **
7940 REM
7950 PRINT#1, "    CPC reflechi ...";
7960 PRINT#1, CHR$(13);
7970 FOR u = 1 TO 3
7980     FOR v = 1 TO 3
7990         essai(u,v) = 0
8000     NEXT v
8010 NEXT u
8020 REM
8030 REM essai autour du dernier jeu
8040 REM
8050 e = INT(3 * RND + 1)
8060 f = INT(3 * RND + 1)
8070 essai(e,f) = 1
8080 e = e - 2
8090 f = f - 2
8100 IF e = 0 AND f = 0 THEN GOTO 8050
8110 x = i + e
8120 y = j + f
8130 flag = 1
8140 IF x < 1 THEN GOTO 8190
8150 IF x > 15 THEN GOTO 8190
8160 IF y < 1 THEN GOTO 8190
8170 IF y > 15 THEN GOTO 8190
8180 GOTO 8200
8190 flag = 0
8200 IF flag = 0 THEN GOTO 8050
8210 IF tableau(x,y) = 46 THEN 8690
8220 flag = 1
8230 FOR u = 1 TO 3
8240     FOR v = 1 TO 3
8250         IF essai(u,v) = 0 THEN flag =
0
8260     NEXT v
8270 NEXT u
8280 IF flag = 0 THEN 8050
```

```
8290 FOR e = -1 TO 1
8300 FOR f = -1 TO 1
8310 IF e = 0 AND f = 0 THEN GOTO 8430
8320 x = i + e
8330 y = j + f
8340 flag = 1
8350 IF x < 1 THEN GOTO 8400
8360 IF x > 15 GOTO 8400
8370 IF y < 1 GOTO 8400
8380 IF y > 15 GOTO 8400
8390 GOTO 8410
8400 flag = 0
8410 IF flag = 0 THEN GOTO 8430
8420 IF tableau(x,y) = 46 THEN 8690
8430 NEXT f
8440 NEXT e
8450 REM
8460 REM ** nouvel essai **
8470 REM
8480 REM cherche une case au hasard
8490 REM
8500 nbessai=0
8510 x = INT(15 * RND + 1)
8520 y = INT(15 * RND + 1)
8530 flag = 1
8540 IF x < 1 THEN GOTO 8590
8550 IF x > 15 GOTO 8590
8560 IF y < 1 GOTO 8590
8570 IF y > 15 GOTO 8590
8580 GOTO 8600
8590 flag = 0
8600 IF flag = 0 THEN 8510
8610 REM
8620 REM ** case trouvee ? **
8630 REM
8640 nbessai = nbessai + 1
8650 PRINT#1, nbessai;
8660 PRINT#1,CHR$(13);
8670 IF nbessai > 90 THEN GOTO 8860
8680 IF tableau(x,y) <> 46 THEN 8510
8690 tableau(x,y) = 233
8700 REM
8710 REM ** affiche position **
8720 REM
8730 PRINT#1, " CPC joue :";x;",";y
8740 LOCATE 3*x,y
8750 PRINT CHR$(tableau(x,y))
```

```
8760 REM
8770 REM ** suite du jeu **
8780 REM
8790 GOTO 7540
8800 IF flag = 0 THEN 8510
8810 GOTO 8480
8820 CLS#1
8830 PRINT#1,"Ok baye baye"
8840 LOCATE 1,22
8850 STOP
8860 CLS #1
8870 PRINT #1, "Ok j'abandonne, vous ete
s trop fort pour moi"
8880 PRINT #1
8890 LOCATE 1,23
8900 STOP
```

Ligne 7000 : initialisation du générateur aléatoire en fonction du temps.

Lignes 7010 à 7290 : rappel des règles et conventions d'entrée des coordonnées.

Lignes 7340 à 7390 : initialisation du tableau représentant le plateau de jeu.

Lignes 7430 à 7510 : affichage du plateau de jeu.

Lignes 7550 à 7570 : le joueur joue premier. Attente des coordonnées.

Ligne 7610 : le joueur abandonne-t-il ? Si oui, exécuter les lignes 8820 à 8850.

Lignes 7620 à 7720 : vérification d'un jeu légal.

Lignes 7800 à 7820 : vérification case vide.

Lignes 7860 à 7910 : le pion est placé.

Lignes 7950 à 8440 : le CPC tente de jouer une position autour du dernier pion posé. Il essaye pour cela aléatoirement une position autour de cette case, jusqu'à ce qu'il ait vérifié toutes les cases.

Lignes 8500 à 8690 : si les essais précédents ont été infructueux, le CPC essaye de trouver une case libre au hasard sur le plateau. Si au bout de 90 essais, il n'a pas trouvé, il abandonne en exécutant les lignes 8860 et suivantes.

Si vous implantez ce programme avec le menu, nous vous conseillons de remplacer les lignes 8850 et 8900 par les lignes suivantes et d'ajouter les lignes 8910 à 8920.

```
8850 GOTO 8900
8900 FOR i% = 1 TO 3000
8910 NEXT i%
8920 RETURN
```

Bien sûr, certains d'entre vous se plaindront peut être que le programme n'est pas suffisamment intelligent, ou encore qu'il ne comptabilise pas automatiquement les alignements réussis, mais nous n'avons pas voulu trop charger ce programme pour permettre au CPC d'accepter d'autres minis jeux pour un même menu.

De plus nous ne doutons pas de vos talents de programmeurs, et comptons sur vous pour améliorer ce programme, ainsi que tous ceux qui vous auront intéressé dans votre revue. Si cela est le cas, nous en serons particulièrement heureux, notre but commence à être atteint.

PILE OU FACE

Mise en situation

Pas de chance ce matin, mal éveillé, vous avez laissé tomber de votre porte-monnaie dix pièces farceuses.

Comble du hasard, toutes les pièces sont tombées côté pile. Tourmenté par ce hasard bizarre, et pris d'un doute financier, vous décidez avant de leur redonner leur place initiale, de toutes les retourner côté face.

Première pièce, pas de problème, elle se retourne docilement. Deuxième pièce, même constatation. Troisième pièce... curieusement, lorsque vous la retournez, la pièce qui la côtoie saute violemment, telle une puce, pour se positionner aussi côté face.

Soit, cela me facilite la tâche pensez-vous. Passons alors à la cinquième pièce. Oh, comble de malchance, voilà la première pièce qui retourne dans son état initial pile.

Pris au jeu, vous reprenez calmement la première pièce qui hélas commande le retournement d'une autre.

Qu'à cela ne tienne, vous reprenez toutes les pièces, et les laissez à nouveau tomber... toutes côté pile encore une fois.

Après plusieurs manœuvres, vous vous apercevez que certaines

pièces retournées provoquent toujours une action sur les mêmes pièces ; vous décidez donc de pousser le jeu en essayant de positionner toutes les pièces côté face en un minimum de retournement.

Et Madame qui attend mon retour du marché ! Tant pis, les provisions seront achetées chez l'épicier du coin, il ferme tard.

Le plateau de jeu sur CPC

Sur votre Amstrad CPC, les dix pièces seront représentées par des étoiles alignées, signalant le côté PILE.

Une flèche, que vous pourrez positionner à l'aide des touches de commande du curseur (flèche gauche et flèche droite) sous une pièce, vous permettra, par l'appui sur la touche <ESPACE> de retourner une pièce.

Le but du jeu est de placer toutes les pièces en position FACE, sachant que certaines pièces réagiront en même temps que d'autres pièces, et ce en un minimum de coups.

Le nombre d'essais sera comptabilisé en haut de l'écran, et lorsque vous aurez placé toutes les pièces dans la position attendue, un menu vous permettra :

- de recommencer le jeu en gardant le même algorithme de retournement des pièces, une même pièce ayant toujours le même effet sur les autres. Ce qui vous permettra d'améliorer votre nombre minimum de tentatives ;
- de recommencer le jeu en utilisant cette fois-ci un algorithme différent de retournement des pièces ;
- de quitter le jeu.

Le listing du programme est le suivant :

```
8960 REM *****
8970 REM ***   PILE   DU   FACE   ***
8980 REM *****
8990 REM
9000 MODE 1
9010 REM
9020 REM *** ANIMATION ***
9030 REM
9040 FOR i% = 1 TO 10
9050 LOCATE 1,8
9060     FOR j% = 1 TO 10
9070         a= INT (RND * 2)
9080         PRINT SPC(1);
9090         IF a = 0 THEN PRINT CHR$(224)
; ELSE PRINT CHR$(238);
9100         PRINT SPC(2)
9110 SOUND 1,2,3,15
9120     NEXT j%
9130 SOUND 2,248,5
9140     PRINT
9150 NEXT i%
9160 REM
9170 LOCATE 10,15
9180 PRINT CHR$(24);
9190 PRINT " PILE DU FACE ";
9200 PRINT CHR$(24);
9210 LOCATE 1,8
9220 FOR j% = 1 TO 10
9230     PRINT SPC(1);
9240     PRINT CHR$(224);
9250     PRINT SPC(2);
9260 NEXT j%
9270 FOR i% = 1 TO 1000
9280 INK 1,0
9290 INK 1,26
9300 NEXT i%
9310 REM
9320 REM *** DEBUT DU JEU ***
9330 REM
9340 MODE 1
9350 hasard = RND * 19
9360 DEG
9370 CLS
9380 LOCATE 1,3
9390 PRINT "Voici la situation de depart
:"
```

```
9400 PRINT
9410 LOCATE 1,15
9420 PRINT "A l'aide des fleches horizon
tales et de"
9430 PRINT "la barre ESPACE, selectionne
z le pion"
9440 PRINT "a retourner"
9450 nombredetourdejeu% = 0
9460 PRINT
9470 LOCATE 1,8
9480 FOR nombre% = 1 TO 10
9490     PRINT SPC(1);
9500     PRINT CHR$(238);
9510     PRINT SPC(2);
9520 NEXT nombre%
9530 PRINT
9540 compteur% = 0
9550 REM
9560 REM REINITIALISATION LIGNE
9570 REM
9580 FOR compteur% = 1 TO 10
9590     position$(compteur%) = CHR$(225)
9600 NEXT compteur%
9610 ancienneposition% = 1
9620 nouvelleposition% = 1
9630 REM
9640 REM *** ATTENTE CHOIX POSITION ***
9650 REM
9660 LOCATE nouvelleposition% * 4 - 2,10
9670 PRINT CHR$(240)
9680 a$ = ""
9690 WHILE a$ <> CHR$(32)
9700     A$ = ""
9710     WHILE a$ = ""
9720         a$ = INKEY$
9730     WEND
9740     IF a$ = CHR$(242) THEN nouvellep
osition% = nouvelleposition% - 1
9750     IF a$ = CHR$(243) THEN nouvellep
osition% = nouvelleposition% + 1
9760     IF nouvelleposition% = 0 THEN no
ouvelleposition% = 10
9770     IF nouvelleposition% = 11 THEN n
ouvelleposition% = 1
9780     LOCATE ancienneposition% * 4 - 2
,10
9790     PRINT CHR$(32)
9800     LOCATE nouvelleposition% * 4 - 2
```

```
,10
9810 PRINT CHR$(240)
9820 ancienneposition% = nouvelleposition%
9830 WEND
9840 REM
9850 REM *** TRAITEMENT DE LA POSITION **
9860 REM
9870 LOCATE nouvelleposition% * 4 - 2,10
9880 PRINT CHR$(32)
9890 REM
9900 REM
9910 IF ancienjeu% = nouvelleposition% THEN GOTO 10100
9920 ancienjeu% = nouvelleposition%
9930 IF position$(nouvelleposition%) = CHR$(224) THEN GOTO 10050
9940 position$(nouvelleposition%) = CHR$(224)
9950 calcul = TAN(hazard + nouvelleposition% / hazard - nouvelleposition%) - SIN(hazard / nouvelleposition%) + 336 * SIN(8 * nouvelleposition%)
9960 calcul = calcul - INT(calcul)
9970 nouvelleposition% = INT(10 * calcul)
9980 REM
9990 REM
10000 IF position$(nouvelleposition%) = CHR$(224) THEN GOTO 10050
10010 position$(nouvelleposition%) = CHR$(224)
10020 GOTO 10270
10030 REM
10040 REM
10050 position$(nouvelleposition%) = CHR$(225)
10060 IF ancienjeu% = nouvelleposition% THEN GOTO 9950
10070 GOTO 10270
10080 REM
10090 REM
10100 IF position$(nouvelleposition%) = CHR$(224) THEN GOTO 10220
10110 position$(nouvelleposition%) = CHR$(224)
```

```
10120 calcul = 0.592 * (1 / TAN(hazard/n
nouvelleposition% + hasard))/SIN(nouvelle
position% * 2 + hasard) - COS(nouvellepo
sition%)
10130 calcul = calcul - INT (calcul)
10140 nouvelleposition% = INT ( 10 * cal
cul)
10150 REM
10160 REM
10170 IF position$(nouvelleposition%) =
CHR$(224) THEN GOTO 10220
10180 position$(nouvelleposition%) = CHR
$(224)
10190 GOTO 10270
10200 REM
10210 REM
10220 position$(nouvelleposition%) = CHR
$(225)
10230 IF ancienjeu% = nouvelleposition%
THEN GOTO 10120
10240 REM
10250 REM *** EFFACEMENT ***
10260 REM
10270 LOCATE 1,3
10280 PRINT SPC(39)
10290 LOCATE 1,15
10300 PRINT SPC(39)
10310 LOCATE 1,16
10320 PRINT SPC(39)
10330 LOCATE 1,17
10340 PRINT SPC(39)
10350 REM
10360 REM *** AFFICHAGE CONFIGURATION NO
UVELLE ***
10370 REM
10380 LOCATE 1,8
10390 FOR nombre% = 1 TO 10
10400 SOUND 1,2,3,15
10410 SOUND 2,248,5
10420 PRINT SPC(1);
10430 IF position$(nombre%) = CHR$(22
5) THEN position$(nombre%) = CHR$(238)
10440 PRINT position$(nombre%);
10450 PRINT SPC(2);
10460 NEXT nombre%
10470 PRINT
10480 REM
10490 REM *** MISE A JOUR ESSAI ***
```

```
10500 REM
10510 nombredetourdejeu% = nombredetourd
ejeu% + 1
10520 LOCATE 30,1
10530 PRINT "Jeu : ";
10540 PRINT nombredetourdejeu%
10550 REM
10560 REM *** TOUT EST FACE ? ***
10570 REM
10580 drapeau% = 0
10590 FOR nombre% = 1 TO 10
10600     IF position$(nombre%) <> CHR$(2
24) THEN drapeau% = 1
10610 NEXT nombre%
10620 IF drapeau% = 1 THEN GOTO 9610
10630 REM
10640 REM nouvelle partie ??
10650 REM
10660 LOCATE 1,16
10670 PRINT "Frappez sur:"
10680 PRINT
10690 PRINT "    1 - garder la meme regl
e"
10700 PRINT "    2 - changer la regle"
10710 PRINT "    3 - terminer le jeu"
10720 a$ = ""
10730 WHILE a$ = ""
10740     a$ = INKEY$
10750 WEND
10760 IF VAL (a$) = 1 THEN GOTO 9370
10770 IF VAL (a$) = 2 THEN GOTO 9350
10780 IF VAL (a$) = 3 THEN RETURN
10790 GOTO 10660
```

Vous pourrez insérer ce programme dans celui des mini-jeux, en remplaçant la ligne 260 par :

```
260 PRINT "4 → PILE OU FACE"
```

La numérotation est ainsi prévue à partir de 8960, pour remplacer les lignes 8980 à 9040 du programme initial de menu.

Lignes 9000 à 9300 : présentation du jeu par une animation simple.

Lignes 9350 : détermination d'un nombre au hasard pour l'algorithme de retournement des pièces.

Lignes 9370 à 9530 : affichage de la position initiale des pièces, côté pile.

Ligne 9540 : remise à zéro du compteur.

Lignes 9580 à 9620 : réinitialisation des paramètres des pièces, et de la flèche.

Lignes 9660 à 9830 : gestion du déplacement de la flèche de choix de la pièce, et acquisition de ce choix.

Lignes 9870 à 10230 : traitement de l'algorithme de retournement, que nous avons voulu suffisamment complexe, pour vous éviter de le découvrir lors de la frappe du listing.

Lignes 10270 à 10340 : effacement de l'ancienne configuration des pièces.

Lignes 10380 à 10470 : affichage de la nouvelle configuration.

Lignes 10510 à 10540 : mise à jour du nombre d'essais effectués.

Lignes 10580 à 10620 : test des pièces pour repérer si elles sont toutes en position face.

Lignes 10660 à fin : présentation du menu pour une nouvelle partie.

LE CHOCOLAT

Mise en situation

« L'un de nous est de trop pour le partager ! », vous êtes-vous écrié en découvrant, en compagnie de votre voisin de palier une fabuleuse malette remplie de pin's introuvables, dans les sous-sols de votre immeuble.

« Je suis entièrement d'accord ! réglons cette affaire à l'amiable, devant une plaque de chocolat », renchérit-il.

Il remarqua, avec amusement, votre surprise, et votre regard effrayé lorsqu'il vous annonça que la plaque de chocolat, qu'il avait déjà sortie de sa poche, contenait un poison fort violent.

Inquiet, vous lui demandez, avant d'accepter de vous proposer les règles du jeu.

Il s'exécuta : « Parmi tous les carrés de cette plaque de chocolat, un seul est empoisonné, que nous positionnerons en haut à gauche.

Chacun notre tour, nous prendrons un morceau de cette plaque, en commençant à partir du bord inférieur droit, sachant que chaque morceau pris ne pourra être que rectangulaire, ou carré, et d'un nombre entier de carré pour chacun de ses côtés.

Celui qui mangera le dernier carré, sera, de force, déclaré perdant, puisqu'il ne sera plus de ce monde. »

Le sourire narquois qu'il eut en terminant sa phrase ne vous impressionna pas, et, fort de toutes vos expériences dans les jeux de réflexion, vous décidâtes que le jeu en valait certainement la chandelle.

Le plateau de jeu sur CPC

La plaque de chocolat proposée par votre voisin est matérialisée sur le CPC par un rectangle composé de symboles carrés, avec en prime le carré supérieur droit représenté par un symbole différent de tous les autres.

Avant de commencer le jeu, vous aurez la possibilité de choisir la taille, en nombre de carrés de côté, de la plaque.

Ensuite, chacun des joueurs (nous avons préféré ne pas vous laisser jouer contre l'ordinateur, ce qui nous aurait certainement conduit à écrire un programme imbattable), choisit la taille du morceau qu'il avalera goulûment à l'aide des touches du curseur et de la barre espace. Le coin supérieur gauche du morceau choisi étant représenté par un symbole différent.

Aura bien sûr perdu, celui qui mangera le morceau empoisonné.

Le listing du programme est le suivant :

```

10960 REM *****
10970 REM ***   LE CHOCOLAT   ***
10980 REM *****
10990 REM
11000 MODE 1
11010 CALL &BB4E
11020 PRINT SPC(10);
11030 PRINT CHR$(24);
11040 PRINT "LE CHOCOLAT EMPOISONNE"
11050 PRINT CHR$(24);
11060 PRINT
11070 PRINT
11080 PRINT "Ce jeu est concu pour deux
joueurs."
11090 PRINT
11100 PRINT "Apres le choix de la taille
d'une plaque";
11110 PRINT "de chocolat, chacun d'entre
-vous doit"
11120 PRINT "en croquer un morceau en s'
efforcant de";
11130 PRINT "laisser a l'adversaire le d
ernier"
11140 PRINT "morceau, qui est empoisonne
."
11150 PRINT
11160 PRINT "Les bons morceaux sont repr
esentes par :";
11170 PRINT SPC(15);
11180 PRINT CHR$(233)
11190 PRINT "Le poison par : "
11200 PRINT SPC(15);
11210 PRINT CHR$(225)
11220 PRINT:PRINT:PRINT:PRINT
11230 PRINT CHR$(24);
11240 PRINT "Frappez une touche ... "
11250 PRINT CHR$(24)
11260 CALL &BBO6
11270 MODE 1
11280 FOR colonne% = 1 TO 10
11290     FOR ligne% = 1 TO 10
11300         chocolat%(ligne%,colonne%) =
0
11310     NEXT ligne%
11320 NEXT colonne%
11330 INPUT "Nom du premier joueur : ";j

```

```
oueur1$
11340 PRINT
11350 INPUT "Nom du deuxieme joueur : ";
joueur2$
11360 PRINT
11370 INPUT "Nombre de ligne (1 a 9) : "
;chocoligne%
11380 IF chocoligne% < 1 OR chocoligne%
> 9 THEN GOTO 11360
11390 PRINT
11400 INPUT "Nombre de colonnes (1 a 9)
: ";chococolonne%
11410 IF chococolonne% < 1 OR chococolon
ne% > 9 THEN GOTO 11390
11420 PRINT
11430 MODE 1
11440 REM
11450 REM *** FABRICATION DU CHOCOLAT **
*
11460 REM
11470 FOR ligne% = 1 TO chocoligne%
11480     FOR colonne% = 1 TO chococolonn
e%
11490         chocolat%(ligne%,colonne%) =
1
11500     NEXT colonne%
11510 NEXT ligne%
11520 REM
11530 REM *** INTRODUCTION DU POISON ***
11540 REM
11550 chocolat%(1,1) = 2
11560 REM
11570 REM *** OUVERTURE DE LA PLAQUE ***
11580 REM
11590 FOR ligne% = 1 TO chocoligne%
11600     LOCATE 15,7 + ligne%
11610     FOR colonne% = 1 TO chococolonn
e%
11620         IF chocolat%(ligne%,colonne%
) = 2 THEN PRINT CHR$(225);
11630         IF chocolat%(ligne%,colonne%
) = 1 THEN PRINT CHR$(233);
11640         IF chocolat%(ligne%,colonne%
) = 0 THEN PRINT CHR$(32);
11650     NEXT colonne%
11660     PRINT
11670 NEXT ligne%
11680 REM
```

```
11690 REM *** CHOIX DU JOUEUR ***
11700 REM
11710 PRINT
11720 LOCATE 1,20
11730 PRINT "Qui commence ?"
11740 LOCATE 16,20
11750 PRINT joueur1$;
11760 LOCATE 30,20
11770 PRINT CHR$(243);" 1"
11780 LOCATE 16,21
11790 PRINT joueur2$;
11800 LOCATE 30,21
11810 PRINT CHR$(243);" 2"
11820 REM
11830 a$ = ""
11840 WHILE a$ <> "1" AND a$ <> "2"
11850     a$ = INKEY$
11860 WEND
11870 LOCATE 1,20
11880 PRINT SPC(39)
11890 LOCATE 1,21
11900 PRINT SPC(39)
11910 REM
11920 REM *** JEU JOUEUR 1 ***
11930 REM
11940 joueur$ = joueur1$
11950 IF a$ = "2" THEN GOTO 12580
11960 a$ = "2"
11970 REM
11980 LOCATE 1,23
11990 PRINT SPC(39)
12000 LOCATE 1,23
12010 PRINT "C'est a : ";
12020 PRINT joueur$
12030 REM
12040 REM *** PASSAGE EN MODE XOR ***
12050 REM
12060 PRINT CHR$(23);
12070 PRINT CHR$(1)
12080 REM
12090 positionligne% = 287
12100 ligne% = 1
12110 colonne% = 1
12120 positioncolonne% = 225
12130 REM
12140 REM *** DEPLACEMENT DU CURSEUR ***
12150 REM
12160 MOVE positioncolonne%,positionlign
```

```
e%
12170 REM
12180 GRAPHICS PEN 1
12190 TAG
12200 PRINT CHR$(231);
12210 MOVE positioncolonne%,positionlign
e%
12220 jeu$ = ""
12230 jeu$ = INKEY$
12240 IF jeu$ = CHR$(240) THEN positionl
igne% = positionligne% + 16 : ligne% = l
igne% - 1
12250 IF jeu$ = CHR$(241) THEN positionl
igne% = positionligne% - 16 : ligne% = l
igne% + 1
12260 IF jeu$ = CHR$(242) THEN positionc
olonne% = positioncolonne% - 16 : colonn
e% = colonne% - 1
12270 IF jeu$ = CHR$(243) THEN positionc
olonne% = positioncolonne% + 16 : colonn
e% = colonne% + 1
12280 IF positionligne% > 287 THEN posit
ionligne% = 287 : ligne% = 1
12290 IF positionligne% < 159 THEN posit
ionligne% = 159 : ligne% = 9
12300 IF positioncolonne% > 353 THEN pos
itioncolonne% = 353 : colonne% = 9
12310 IF positioncolonne% < 225 THEN pos
itioncolonne% = 225 : colonne% = 1
12320 PRINT CHR$(231);
12330 IF jeu$ <> CHR$(32) THEN GOTO 1216
0
12340 REM
12350 REM *** TRAITEMENT ***
12360 REM
12370 TAGOFF
12380 IF chocolat%(ligne%,colonne%) = 2
THEN GOTO 12640
12390 IF chocolat%(ligne%,colonne%) = 0
THEN GOTO 12160
12400 FOR comptageligne% = ligne% TO cho
cocolonne%
12410     FOR comptagecolonne% = colonne%
TO chococolonne%
12420         chocolat%(comptageligne%,com
ptagecolonne%) = 0
12430     NEXT comptagecolonne%
12440 NEXT comptageligne%
```

```
12450 FOR ligne% = 1 TO chocoligne%
12460     LOCATE 15,7 + ligne%
12470     FOR colonne% = 1 TO chococolonn
12480         IF chocolat%(ligne%,colonne%
) = 2 THEN PRINT CHR$(225);
12490         IF chocolat%(ligne%,colonne%
) = 1 THEN PRINT CHR$(233);
12500         IF chocolat%(ligne%,colonne%
) = 0 THEN PRINT CHR$(32);
12510     NEXT colonne%
12520     PRINT
12530 NEXT ligne%
12540 GOTO 11940
12550 REM
12560 REM *** JEU JOUEUR 2 ***
12570 REM
12580 joueur$ = joueur2$
12590 a$ = "1"
12600 GOTO 11980
12610 REM
12620 REM *** UN JOUEUR A PERDU ***
12630 REM
12640 MODE 0
12650 FOR i% = 1 TO 4
12660     PRINT CHR$(24)
12670     CLS
12680 NEXT i%
12690 FOR i% = 1 TO 6
12700     PRINT CHR$(24)
12710     LOCATE 1,10
12720     PRINT " AHHHhhhhhhrrrgg !!!"
12730     CLS
12740 NEXT i%
12750 MODE 1
12760 LOCATE 1,10
12770 PRINT joueur$;
12780 PRINT " a perdu ..."
12790 PRINT:PRINT:PRINT
12800 PRINT "Une autre partie (O/N) ?"
12810 WHILE a$ <> "O" AND a$ <> "N"
12820     a$ = INKEY$
12830     a$ = UPPER$(a$)
12840 WEND
12850 IF a$ = "O" THEN GOTO 11360
12860 RETURN
```

Ce programme pourra être inséré en modifiant la ligne 280 du menu des mini-jeux par :

280 PRINT "5 → LE CHOCOLAT"

Il sera ainsi positionné à partir de la ligne 10960, qui remplaceront les lignes 10980 à 11040.

Lignes 11000 à 11320 : présentation du jeu.

Lignes 11330 à 11360 : acquisition des noms des joueurs.

Lignes 11370 à 11420 : acquisition de la taille de la plaque de chocolat.

Lignes 11450 à 11550 : initialisation des carrés de la plaque de chocolat, en n'oubliant pas le carré empoisonné.

Lignes 11590 à 11670 : affichage de la plaque.

Lignes 11710 à 11900 : on laisse le choix du premier jeu au joueur qui le désire.

Lignes 11980 à 12020 : affichage du tour de jeu du joueur concerné.

Lignes 12060 à 12120 : initialisation de la position initiale du curseur et du mode d'affichage XOR, permettant un affichage en surimpression.

Lignes 12160 à 12330 : gestion en mode graphique du déplacement du curseur.

Lignes 12370 à 12540 : remise à jour de la plaque de chocolat, et affichage des carrés restant.

Lignes 12580 à 12600 : initialisation pour le joueur suivant.

Lignes 12640 à 12790 : animation à l'écran pour le joueur qui mange le morceau empoisonné.

Lignes 12800 à 12860 : proposition d'une autre partie.

LE MONSTRE DU LOCH NESS

Un peu d'histoire

C'est autour de l'année 1933 que fut aperçue une forme étrange, semblable au long cou d'un animal dans le lac Ness, en anglo-saxon Loch Ness, situé dans la région des Highlands de la brumeuse Ecosse au sud-ouest de la ville d'Inverness.

Une soirée bien arrosée ? L'atmosphère mystérieuse des lieux ? Ou la facétie de certains promeneurs désirant se faire connaître ? Toujours est-il qu'une étrange forme, semblable à un animal venu des fonds des temps est aperçue de temps à autres.

Quelques photos un peu vieilles, dont il sera impossible de déterminer s'il y a trucage ou pas, attisent la rumeur, et font ainsi naître la légende (ou pas) de Nessie (dont le nom scientifique est, depuis 1972, *Nessiteras rhombopteryx*), le monstre du Loch Ness.

Pèlerinage et activité commerciale se développent autour de ce brave animal que peu de personne reverront.

Des expéditions scientifiques de grandes envergures furent entreprises pour déloger le monstre de sa tanière.

Certains scientifiques affirment tout de même son existence en parlant d'une espèce animale de notre ère qui aurait pu, par mutation probablement, atteindre une taille hors du commun.

Phénomène toujours étrange, puisque tout corps humain ou animal, tombant dans le Loch Ness, profond de 300 mètres, est en général disparu à jamais, aucun reste n'y étant jamais retrouvé.

Et votre rôle dans tout cela ?

Las de leurs recherches au demeurant fort coûteuses, les scientifiques vous ont confié une barque, un filet et un sonar perfectionné (bâti autour de l'unité centrale d'un CPC) pour retrouver ce monstre.

Le plateau de jeu

Considérant que le monstre fut dernièrement repéré dans une zone approximativement cubique de vingt miles de côté, pour une profondeur de dix miles, vous déciderez d'une position de la barque à l'aide des coordonnées (les extrêmes étant rappelées à l'écran).

Vous jetterez ensuite votre filet en indiquant la profondeur de pêche.

Le sonar vous indiquera ensuite la position de votre filet par rapport au monstre en utilisant les quatre points cardinaux (nord, est, ouest ou sud) et sa profondeur plus ou moins suffisante.

Attention, vous n'aurez qu'un nombre limité d'essais, car toute cette agitation de l'eau réveillera probablement le monstre, qui ne fera qu'une bouchée de votre coquille de noix.

Le listing du programme est le suivant :

```
12950 REM *****
12960 REM *** LA CHASSE AU MONSTRE ***
12970 REM ***   DU LOCH   NESS   ***
12980 REM *****
12990 REM
13000 CALL &BB4E
13010 MODE 1
13020 MOVE 200,250
13030 DRAW 350,350
13040 DRAW 550,350
13050 DRAW 420,250
13060 DRAW 200,250
13070 DRAW 200,150
13080 DRAW 420,150
13090 DRAW 420,250
13100 MOVE 550,350
13110 DRAW 550,250
13120 DRAW 420,150
13130 TAG
13140 MOVE 130,250
13150 PRINT "0,0,0";
13160 MOVE 420,250
13170 PRINT "20,0,0";
13180 MOVE 530,368
13190 PRINT "20,20,0";
13200 MOVE 420,150
13210 PRINT "20,0,20";
13220 MOVE 270,230
13230 PRINT "LOCH";
13240 MOVE 270,190
13250 PRINT "NESS";
13260 MOVE 400,376
13270 PRINT "NORD";
13280 TAGOFF
13290 WINDOW #5,1,40,40,20
13300 PRINT #5,CHR$(24)
13310 CLS #5
13320 REM
13330 REM *** PRESENTATION ***
13340 REM
13350 PRINT #5,"Vous avez ete designe po
ur attraper le"
13360 PRINT #5,"monstre du Loch'Ness, po
urvu d'une"
13370 PRINT #5,"barque et d'un filet que
vous lancerez"
```

```
13380 PRINT #5,"dans le lac. Vous etes e
quipe d'un sonar";
13390 PRINT #5,"qui vous aidera dans vos
recherches.
13400 PRINT #5,"avant que le monstre ne
vous croque."
13410 REM
13420 CALL &BB06
13430 REM
13440 REM *** DIMENSIONS DU LAC ***
13450 REM
13460 taillexlac% = 20
13470 tailleylac% = 20
13480 profondeurlac% = 10
13490 REM
13500 REM *** NOMBRE D'ESSAIS ***
13510 REM
13520 nombredessai% = taillexlac% * tail
leylac% * profondeurlac% / 500
13530 REM
13540 REM *** EMPLACEMENT DU MONSTRE ***
13550 REM
13560 coordonneexmonstre% = INT ( taille
xlac% * RND)
13570 coordonneeymonstre% = INT ( taille
ylac% * RND)
13580 profondeurmonstre% = INT (profonde
urlac% * RND)
13590 REM
13600 CLS #5
13610 REM
13620 REM *** ESSAIS SUCCESSIFS ***
13630 REM
13640 FOR essai% = 1 TO nombredessai%
13650     LOCATE #0,1,1
13660     PRINT CHR$(24);
13670     PRINT "Essai numero";
13680     PRINT essai%;
13690     PRINT CHR$(24)
13700     INPUT #5, "emplacement de la ba
rque (x,y): ";emplacementx%,emplacementy
%
13710     INPUT #5, "profondeur du filet:
";profondeur%
13720     IF (emplacementx% = coordonneex
monstre%) AND (emplacementy% = coordonne
eymonstre%) AND (profondeur% = profondeu
rmonstre%) THEN GOTO 14190
```

```
13730     REM
13740     REM *** RECHERCHE AU SONAR ***
13750     REM
13760     position$ = ""
13770     PRINT #5
13780     PRINT #5, "D'apres le sonar, le
filet se trouve"
13790     PRINT #5, "      ";
13800     IF emplacementy% > coordonneeym
onstre% THEN position$ = position$ + "au
Nord"
13810     IF emplacementy% < coordonneeym
onstre% THEN position$ = position$ + "au
Sud"
13820     IF emplacementx% > coordonneexm
onstre% THEN IF ((MID$(position$,4,1) =
"N") OR (MID$(position$,4,1) = "S")) THE
N position$ = position$ + "-Est" ELSE po
sition$ = position$ + " a l'Est"
13830     IF emplacementx% < coordonneexm
onstre% THEN IF ((MID$(position$,4,1) =
"N") OR (MID$(position$,4,1) = "S")) THE
N position$ = position$ + "-Ouest" ELSE
position$ = position$ + " a l'Ouest"
13840     IF (emplacementx% <> coordonnee
xmonstre% OR emplacementy% <> coordonnee
ymonstre%) THEN position$ = position$ +
" et"
13850     REM
13860     REM profondeur
13870     REM
13880     IF profondeur% = profondeurmons
tre% THEN position$ = position$ + " a la
bonne profondeur"
13890     IF profondeur% > profondeurmons
tre% THEN position$ = position$ + " trop
profond"
13900     IF profondeur% < profondeurmons
tre% THEN position$ = position$ + " pas
assez profond"
13910     REM
13920     REM indication
13930     REM
13940     PRINT #5, position$
13950     PRINT #5 : PRINT #5
13960 NEXT essai%
13970 REM
13980 REM *** PLUS D'ESSAIS ***
```

```

13990 REM
14000 FOR i% = 1 TO 30
14010     PRINT #5 , CHR$(24)
14020     CLS #5
14030 NEXT i%
14040 PRINT #5 : PRINT #5 : PRINT #5 : P
RINT #5
14050 PRINT #5, "Desole, le monstre vien
t de remonter"
14060 PRINT #5, "et a coule votre barque
"
14070 PRINT #5
14080 PRINT #5, "Une autre partie ? (O/N
)"
14090 a$ = ""
14100 WHILE a$ = ""
14110     a$ = INKEY$
14120     a$ = UPPER$(a$)
14130 WEND
14140 IF a$ = "O" THEN GOTO 13000
14150 RETURN
14160 REM
14170 REM *** LE MONSTRE EST PECHE ***
14180 REM
14190 CLS #5
14200 PRINT #5 : PRINT #5 : PRINT #5 : P
RINT #5
14210 PRINT #5, "BRAVO vous etes le prem
ier a avoir"
14220 PRINT #5, "decouvert le monstre du
Loch Ness"
14230 GOTO 14070
14240 END

```

Il n'a pas été prévu dans le menu des mini-jeux l'appel de ce jeu, mais vous pourrez sans problème l'ajouter grâce aux lignes suivantes :

```
281 LOCATE 25,15
```

```
282 PRINT "6 → NESSIE"
```

```
330 WHILE a$ <> "1" AND a$ <> "2" AND a$ <> "3"
      AND a$ <> "4" AND a$ <> "5" AND aa$ <> "6"
```

```
580 ON a$ GOSUB 3000, 5000, 7000, 9000, 11000, 13000
```

Le programme se placera à partir de la ligne 12950.

Lignes 13000 à 13310 : dessin de la zone des recherches avec affichage des coordonnées.

Lignes 13350 à 13420 : présentation de la mission scientifique.

Lignes 13460 à 13480 : initialisation des dimensions du lac, que vous pourrez modifier selon votre expérience de la pêche.

Ligne 13520 : calcul du nombre d'essais maximal en fonction des dimensions du lac.

Lignes 13560 à 13580 : positionnement aléatoire du monstre.

Lignes 13640 à 13710 : acquisition des différents essais.

Ligne 13720 : vérification si essai fructueux.

Lignes 13760 à 13840 : recherche au sonar sur la position horizontale.

Lignes 13880 à 13900 : recherche au sonar sur la profondeur.

Lignes 13940 et 13950 : affichage de la recherche au sonar.

Lignes 14000 à 14070 : le pêcheur a perdu.

Lignes 14080 à 14150 : proposition d'une autre partie.

Lignes 14190 à fin : traitement en cas de réussite.